
IEEE Method – Transient or Steady State Calculation of Bare Overhead Conductor Temperature or Thermal Rating
Initialize Variables and Arrays
Select Desired Calculation
Steady-State Conductor Temperature Given Current
Ampacity (Steady-State Thermal Rating)
Conductor Temperature Given Step Change in Current
Transient Thermal or Fault Rating
YES
Enter Input Data from File
NO
Load Data from File
Subroutine to Enter Input Data
Subroutine to Calculate Conductor Solar Heat Gain (QS)
Subroutine to Calculate Therm Coef of RAC & Heatcap & Wind Correction
Steady-State Conductor Temperature Given Current
Ampacity (Steady-State Thermal Rating)
Conductor Temperature Given Step Change in Current
Transient Thermal or Fault Rating
Iterate on Steady-state Current to find Conductor Temp
Calc Ampacity Given Conductor Temp & Weather Conditions
Iterate on current steps to track Cond Temp for Step-change in Current
Iterate on current steps to track Cond Temp & repeat until Tc = Tmax
Print Output to Screen
YES
Rerun?
NO
END

Program listing

```
10 REM ****************************************************************
20 REM *   IEEE METHOD - TRANSIENT OR STEADY STATE CALCULATION
30 REM *  OF BARE OVERHEAD CONDUCTOR TEMPERATURE OR THERMAL RATING
40 REM *
50 REM *
60 REM *
70 REM *    ASSUMES SI UNITS FOR INPUT
80 REM *
90 REM *
240 REM * IN COMPARISON WITH THE 1986 VERSION OF THIS PROGRAM, PROVIDED
250 REM * BY THE IEEE, THE 1993 VERSION ADDED THE FOLLOWING FEATURES:
260 REM *
290 REM *  - INITIAL CONDUCTOR TEMP OR CURRENT CAN BE USED IN
300 REM *    TRANSIENT CALCULATIONS
330 REM *  - VERY SHORT DURATION "FAULT" CURRENTS AS LARGE AS 1E6
340 REM *    AMPERES FOR TIMES AS SHORT AS 0.01 SEC CAN BE USED
350 REM *  - THE ORIGINAL NUMERICAL ITERATION METHOD HAS BEEN
360 REM *    REPLACED WITH A MUCH MORE EFFICIENT METHOD
370 REM *  - FOR ACSR CONDUCTOR, THE HEAT CAPACITY OF THE STEEL CORE
380 REM *    AND THE OUTER ALUM STRANDS ARE ENTERED SEPARATELY.
390 REM *
392 REM *   THIS VERSION IS CONSISTENT WITH IEEE 738-2012
394 REM *  - THE SOLAR MODEL ALLOWS ANY HOUR AND LATITUDE
396 REM *  - THE AIR PROPERTIES ARE CALCULATED WITH CLOSED FORM EQUATIONS
398 REM *  - THIS PROGRAM AND EQUATIONS USE SI UNITS
400 REM *************************************************************
410 REM ***********************************
420 REM * INITIALIZE VARIABLES AND ARRAYS *
430 REM ***********************************
440 DIM ATCDR(1000)
450 DIM TIME(1000)
460 FLAG1 = 0
470 XIDUMMY = 0
480 XIPRELOAD = 0
490 XISTEP = 0
500 TCDR = 0
510 TCDRPRELOAD = 0
520 TCDRMAX = 0
530 IORTPRELOAD = 0
540 DELTIME = 0
550 FS1 = 0
560 FS2 = 0
570 FS3 = 0
580 X$ = STRING$(56, 45)
590 REM *******************************
600 REM * START REPEAT CALCULATION HERE
610 REM *******************************
620 FOR KI = 1 TO 1000
630 ATCDR(KI) = 0
640 TIME(KI) = 0
650 NEXT KI
660 NFLAG = 0
670 PI = 3.141593
672 PIANG = PI / 180!
680 IF FLAG1 = 99 GOTO 1120
690 REM ***********************************************
700 REM * SPECIFY DATA INPUT ASCII FILE NAME
710 REM ***********************************************
720 CLS
730 INPUT "ENTER INPUT FILE NAME ", F$: OPEN F$ FOR INPUT AS #1
850 REM ***********************************************************
860 REM * ENTER DATA FROM INPUT FILE
```

```
870 REM ************************************************************
880 GOSUB 8000
1120 REM *****************************************
1130 REM * CALCULATE SOLAR HEAT INPUT TO CONDUCTOR
1140 REM *****************************************
1150 GOSUB 5000
1160 REM **************************************************************
1170 REM * CALCULATE THERMAL COEF OF RESISTANCE & WIND ANGLE CORRECTION
1180 REM **************************************************************
1190 GOSUB 9000
1200 REM ********************************
1210 REM * SELECT THE CALCULATION DESIRED
1220 REM ********************************
1230 ON NSELECT GOTO 1500, 1240, 1460, 1460
1240 REM ********************************************************************
1250 REM *          FOR NSELECT = 2
1260 REM * GO TO AMPACITY SUBROUTINE TO CALCULATE THE STEADY STATE
1270 REM * CURRENT (TR) GIVEN THE STEADY STATE CONDUCTOR TEMPERATURE (TCDR)
1280 REM * THE CONDUCTOR TEMPERATURE IS GIVEN SO ONLY ONE PASS THROUGH
1290 REM * THE SUBROUTINE IS REQUIRED.
1300 REM ********************************************************************
1310 TCDR = TCDRPRELOAD
1320 GOSUB 15000
1330 REM *****************************
1340 REM ********************************************************************
1350 REM
1360 REM ********************************************************************
1370 REM *          FOR NSELECT = 1,3,OR 4
1380 REM * GO TO AMPACITY SUBROUTINE REPEATEDLY IN ORDER TO CALCULATE
1390 REM * THE STEADY STATE CURRENT (TR) CORRESPONDING TO TRIAL VALUES OF
1400 REM * CONDUCTOR TEMPERATURE (TCDR). IF T=1 THEN THE OUTPUT OF THE
1410 REM * SUBROUTINE, TR, IS THE STEADY STATE CURRENT FOR
1420 REM * WHICH A STEADY STATE TEMPERATURE WAS TO BE FOUND.
1430 REM * IF T=3 OR 4 AND IORTPRELOAD=1, THEN TR IS THE INITIAL PRE-STEP
1440 REM * CHANGE CURRENT FOR WHICH AN INITIAL TEMPERATURE WAS TO BE
CALCULATED.
1450 REM *********************************************************************
1460 ON IORTPRELOAD GOTO 1500, 1650
1470 REM ********************************************
1480 REM * CALCULATE TCDR GIVEN XIDUMMY = XIPRELOAD *
1490 REM ********************************************
1500 XIDUMMY = XIPRELOAD
1510 NFLAG = 0
1520 GOSUB 13000
1530 TCDRPRELOAD = TCDR
1540 REM ***************************************************************
1550 REM * FOR NSELECT = 1 THE PROGRAM HAS FOUND THE STEADY STATE CONDUCTOR
1560 REM * TEMPERATURE (TCDRPRELOAD) CORRESPONDING TO THE GIVEN STEADY STATE
1570 REM * CURRENT (XIPRELOAD)
1580 REM ***************************************************************
1590 IF NSELECT = 1 THEN 1730
1600 REM *****************************************************************
1610 REM * FOR NSELECT = 3 OR 4, THE PROGRAM HAS DETERMINED (IORTPRELOAD=1) OR
BEEN
1620 REM * GIVEN (IORTPRELOAD=2) THE INITIAL STEADY STATE CONDUCTOR TEMPERATURE
1630 REM * AND CONTROL PASSES TO FURTHER TRANSIENT CALCULATIONS
1640 REM *****************************************************************
1650 IF NSELECT = 4 THEN GOSUB 10000
1660 REM *************************************************************
1670 REM * BEGIN CALCULATION OF CONDUCTOR TEMP AS A FUNCTION OF TIME
1680 REM * FOR A STEP INCREASE IN ELECTRICAL CURRENT, NSELECT = 3
1690 REM *************************************************************
1700 ET = 3600!
```

```
1710 XISTEP = XISTEP
1720 GOSUB 11000
5000 REM ////////////////////////////////////////////////////////////
5010 REM / SUBROUTINE TO CALCULATE CONDUCTOR SOLAR HEAT GAIN (QS)
5020 REM ////////////////////////////////////////////////////////////
5030 IF SUN.TIME >= 24 THEN 5560
5040 DEG.TO.RAD = PI / 180!
5050 CDR.LAT.RAD = CDR.LAT.DEG * DEG.TO.RAD
5060 REM * SOLAR DECLINATION
5070 DECL.DEG = 23.4583 * SIN(((284 + NDAY) / 365) * 2 * PI)
5080 DECL.RAD = DECL.DEG * DEG.TO.RAD
5090 REM * SOLAR ANGLE RELATIVE TO NOON
5100 HOUR.ANG.DEG = (SUN.TIME - 12) * 15
5110 HOUR.ANG.RAD = HOUR.ANG.DEG * DEG.TO.RAD
5120 REM * FIND SOLAR ALTITUDE - H3
5130 H3ARG = COS(CDR.LAT.RAD) * COS(DECL.RAD) * COS(HOUR.ANG.RAD) +
SIN(CDR.LAT.RAD) * SIN(DECL.RAD)
5140 H3.RAD = ATN(H3ARG / SQR(1 - H3ARG ^ 2))
5150 H3.DEG = H3.RAD / DEG.TO.RAD
5160
5170 IF A3 = 1 THEN 5290
5180 REM ***************************************************************
5190 REM * SOLAR HEATING (Q3) AT EARTH SURFACE (W/M2) IN CLEAR AIR (P6)
5200 REM ***************************************************************
5210 Q3 = -42.2391 + 63.8044 * H3.DEG - 1.922 * H3.DEG ^ 2
5220 Q3 = Q3 + .034692 * H3.DEG ^ 3 - 3.6112E-04 * H3.DEG ^ 4
5230 Q3 = Q3 + 1.9432E-06 * H3.DEG ^ 5 - 4.0761E-09 * H3.DEG ^ 6
5240 B$ = "CLEAR"
5250 GOTO 5330
5260 REM ****************************************************************
5270 REM * SOLAR HEAT (Q3) AT EARTH SURFACE (W/M2) IN INDUSTRIAL AIR (P6)
5280 REM ****************************************************************
5290 Q3 = 53.1821 + 14.211 * H3.DEG + .66138 * H3.DEG ^ 2
5300 Q3 = Q3 - .031658 * H3.DEG ^ 3 + 5.4654E-04 * H3.DEG ^ 4
5310 Q3 = Q3 - 4.3446E-06 * H3.DEG ^ 5 + 1.3236E-08 * H3.DEG ^ 6
5320 B$ = "INDUSTRIAL"
5330 REM * CALCULATE SOLAR AZIMUTH VARIABLE, CHI
5335
5340 CHI.DENOM = SIN(CDR.LAT.RAD) * COS(HOUR.ANG.RAD) - COS(CDR.LAT.RAD) *
TAN(DECL.RAD)
5350 CHI = SIN(HOUR.ANG.RAD) / CHI.DENOM
5360 REM * CALCULATE SOLAR AZIMUTH CONSTANT, CAZ
5370 IF HOUR.ANG.DEG < 0 AND CHI >= 0 THEN
   CAZ = 0
5380 ELSEIF HOUR.ANG.DEG >= 0 AND CHI < 0 THEN
   CAZ = 360
5390 ELSE
   CAZ = 180
5495 END IF
5400 REM * CALCULATE SOLAR AZIMUTH IN DEGREES, Z4.DEG
5410 Z4.DEG = CAZ + ATN(CHI)
5420 Z4.RAD = Z4.DEG * DEG.TO.RAD
5510 Z1.RAD = Z1.DEG * DEG.TO.RAD
5520 E1 = COS(H3.RAD) * COS(Z4.RAD - Z1.RAD)
5530 E2.RAD = ATN(SQR(1 / E1 ^ 2 - 1))
5540 QS = ABSORP * Q3 * SIN(E2.RAD) * D / 1000 * (1 + .0001148 * CDR.ELEV -
1.108E-08 * CDR.ELEV ^ 2)
5542 IF QS < 0 THEN QS=0.0
5545
5550 GOTO 5570
5560 QS = 0!
5570 RETURN
8000 REM //////////////////////////////////
```

```
8010 REM / SUBROUTINE TO ENTER INPUT DATA
8020 REM //////////////////////////////////
8030 REM NSELECT IS TYPE OF CALCULATION
8040 REM 1 = STEADY-STATE TEMP, 2 = STEADY-STATE RATING
8045 REM 3 = TRANSIENT TEMP, 4 TRANSIENT RATING
8150 REM *****************
8160 REM * TRANSIENT DATA
8170 REM *****************
8180 INPUT #1, IORTPRELOAD, Z$
8190 IF IORTPRELOAD = 1 THEN INPUT #1, XIPRELOAD, Z$
8200 IF IORTPRELOAD = 2 THEN INPUT #1, TCDRPRELOAD, Z$
8210 IF NSELECT = 4 THEN INPUT #1, TCDRMAX, Z$ ELSE TCDRMAX = 1000
8220 IF NSELECT = 3 THEN INPUT #1, XISTEP, Z$
8230 INPUT #1, SORM, Z$
8240 INPUT #1, TT, Z$
8250 INPUT #1, DELTIME, Z$
8260 IF SORM = 1 THEN TT = TT * 60
8270 REM **************
8280 REM * WEATHER DATA
8290 REM **************
8300 INPUT #1, TAMB, Z$
8310 INPUT #1, VWIND, Z$
8320 INPUT #1, WINDANG.DEG, Z$
8340 REM *****************
8350 REM * CONDUCTOR DATA
8360 REM *****************
8370 INPUT #1, C$, Z$
8380 INPUT #1, D, Z$
8390 INPUT #1, TLO, THI, Z$
8400 INPUT #1, RLO, RHI, Z$
8430 RLO = RLO / 1000
8440 RHI = RHI / 1000
8450 IF NSELECT = 1 OR NSELECT = 2 THEN 8510
8460 INPUT #1, HNH, Z$
8470 IF HNH = 1 THEN INPUT #1, HEATOUT, Z$: HEATCORE = 0
8480 IF HNH = 2 THEN INPUT #1, HEATOUT, HEATCORE, Z$
8490 REM
8500 REM
8510 HEATCAP = HEATOUT + HEATCORE
8520 INPUT #1, EMISS, ABSORP, Z$
8530 INPUT #1, CDR.ELEV, Z$
8540 INPUT #1, Z1.DEG, Z$
8550 REM
8560 REM ********************
8570 REM * SOLAR HEATING DATA
8580 REM ********************
8585 REM SPECIFY LATITUDE AND SUN TIME
8590 INPUT #1, CDR.LAT.DEG, Z$
8600 INPUT #1, SUN.TIME, NDAY, Z$
8610 INPUT #1, A3, B$, Z$
8620 RETURN
9000 REM
///////////////////////////////////////////////////////////////////////
9010 REM / SUBROUTINE TO CALCULATE THERM COEF OF RAC & HEATCAP & WIND
CORRECTION
9020 REM
///////////////////////////////////////////////////////////////////////
9030 REM ***********************************************************
9040 REM * SETUP LINEAR CONDUCTOR RESISTANCE EQ AS FUNCTION OF TEMP
9042 REM * B IN OHM/M-C AND B1 IN OHM/M
9050 REM ***********************************************************
9060 B = (RHI - RLO) / (THI - TLO)
9070 B1 = RLO - B * TLO
```

```
9080 REM ****************************************************
9090 REM * SET UP LINEAR HEAT CAPACITY EQS AS FUNCTION OF TEMP
9100 REM ****************************************************
9110 REM **************************************************
9120 REM * CORRECTION FACTOR (YC) FOR NON-PERPENDICULAR WIND
9130 REM **************************************************
9140 WINDANG.RAD = 1.570796 - WINDANG.DEG * PIANG
9150 YC = 1.194 - SIN(WINDANG.RAD) - .194 * COS(2! * WINDANG.RAD) + .368 *
SIN(2! * WINDANG.RAD)
9160 RETURN
10000 REM ///////////////////////////////////////////////////////////////////
10010 REM / SUBROUTINE TO CALCULATE STARTING VALUE FOR CURRENT ITERATION
10020 REM / BY ASSUMING ADIABATIC HEATING DURING TIME TT
10030 REM ///////////////////////////////////////////////////////////////////
10040 TCDR = (TCDRMAX + TAMB) / 2
10050 IF TT < 60 THEN HEATCAP = HEATOUT ELSE HEATCAP = HEATOUT + HEATCORE
10060 GOSUB 15000
10070 AT = SQR(HEATCAP * (TCDRMAX - TAMB) / TT) / W4
10080 TCDR = TCDRPRELOAD
10090 NFLAG = 1
10100 GOSUB 13000
10110 RETURN
11000 REM //////////////////////////////////////////////////////////////
11010 REM / SUBROUTINE CALCS CDR TEMP VS TIME FOR STEP CHANGE CURRENT
11020 REM //////////////////////////////////////////////////////////////
11030 IF NSELECT = 4 THEN PRINT USING "TRYING A CURRENT OF #######.#### AMPS";
XISTEP
11040 FLAG = 0
11050 ATCDR(1) = TCDRPRELOAD
11060 TCDR = ATCDR(1)
11070 GOSUB 15000
11080 K = 1
11090 ATCDR(K + 1) = TCDR + (W4 ^ 2 * XISTEP ^ 2 + QS - QR - QC) * DELTIME /
HEATCAP
11100 TIME(K + 1) = TIME(K) + DELTIME
11110 TCDR = ATCDR(K + 1)
11115 IF NSELECT = 4 GOTO 11130
11120 PRINT "TIME = "; TIME(K + 1); " SECONDS / "; "CDR TEMP = "; TCDR; "DEG C"
11130 IF NSELECT = 3 AND TCDR > TCDRMAX THEN 11280
11140 REM *****************************************************************
11150 REM *
11160 REM *****************************************************************
11170 GOSUB 15000
11180 K = K + 1
11190 IF K = 3000 THEN PRINT "TIME INTERVAL TOO SMALL. ARRAY OUT OF BOUNDS ":
GOTO 1880
11200 IF TIME(K) < TT THEN 11090
11210 IF XISTEP = 0 AND TCDR > TCDRMAX THEN 11220 ELSE 11250
11220 PRINT "EVEN IF THE CURRENT IS REDUCED TO ZERO AMPS, THE CONDUCTOR"
11230 PRINT USING "TEMPERATURE WILL NOT DECREASE TO ####.# DEG C IN ####.#
MINUTES"; TCDRMAX; TT / 60
11240 GOTO 1880
11250 REM **********************************
11260 REM * CHECK FOR SHORT DURATION FAULTS
11270 REM **********************************
11280 IF TIME(K) >= 60 OR FLAG = 1 OR HEATCORE = 0 OR TT < 60 THEN GOTO 11320
11290 HEATCAP = HEATOUT
11300 FLAG = 1
11310 GOTO 11050
11320 KTIMEMAX = K
11330 RETURN
12000 REM //////////////////////////////////////////////////////
12010 REM / SUBROUTINE ITERATES TO FIND CONDUCTOR TEMPERATURE
```

```
12020 REM / GIVEN THE CONDUCTOR CURRENT
12030 REM ////////////////////////////////////////////////////
12040 IF NFLAG = 0 THEN TCDR = X: GOSUB 15000: TEMP = XIDUMMY - TR: RETURN
12050 IF NFLAG = 1 THEN XISTEP = X: GOSUB 11000
12060 IF TCDRPRELOAD <= TCDRMAX THEN TEMP = TCDRMAX - TCDR: RETURN
12070 IF TCDRPRELOAD > TCDRMAX THEN TEMP = TCDR - TCDRMAX: RETURN
13000 REM ////////////////////////////////////////////////////////////////////
13010 REM / SUBROUTINE RTMI MUELLER-S ITERATION METHOD SELECTS A CURRENT  /
13020 REM / WHICH JUST RAISES TCDR TO TCDMAX IN THE TIME TT. THIS CURRENT /
13030 REM / IS THE TRANSIENT RATING OF THE CONDUCTOR. IT DOES THIS BY    /
13040 REM / REPEATEDLY GUESSING A CURRENT - XISTEP - CALCULATING TCDR AT TT /
13050 REM / AND COMPARING THE CALCULATED TCDR TO TCDRMAX. ROUTINE SUPPLIED /
13060 REM / COURTESY OF BILL HOWINGTON.                            /
13070 REM ///////////////////////////////////////////////////////////////////
13080 REM * START BY PREPARING TO ITERATE
13090 REM ******************************
13100 XLI = 0: XRI = 0: EPS = .049: IEND = 20: X = 0
13110 GOSUB 14000
13120 IER = 0: XL = XLI: XR = XRI: X = XL: TOL = X
13130 GOSUB 12000
13140 F = TEMP: IF XLI = XRI OR F = 0 THEN 13530
13150 FL = F: X = XR: TOL = X
13160 GOSUB 12000
13170 F = TEMP: IF F = 0 THEN 13530
13180 FR = F: IF (SGN(FL) + SGN(FR)) = 0 THEN 13200 ELSE 13760
13190 REM ***********************************************
13200 REM BASIC ASSUMPTION FL*FR LESS THAN 0 IS SATISFIED.
13210 REM ***********************************************
13220 I = 0
13230 REM *******************
13240 REM START ITERATION LOOP
13250 REM *******************
13260 I = I + 1
13270 REM *******************
13280 REM START BISECTION LOOP
13290 REM *******************
13300 FOR JK = 1 TO IEND
13310 X = .5 * (XL + XR): TOL = X: GOSUB 12000
13320 F = TEMP: IF F = 0 THEN 13530
13330 IF (SGN(F) + SGN(FR)) = 0 THEN 13370 ELSE 13380
13340 REM **********************************************************
13350 REM INTERCHANGE XL AND XR IN ORDER TO GET THE SAME SIGN IN F AND FR
13360 REM **********************************************************
13370 TOL = XL: XL = XR: XR = TOL: TOL = FL: FL = FR: FR = TOL
13380 TOL = F - FL: DA = F * TOL: DA = DA + DA
13390 IF (DA - FR * (FR - FL)) >= 0 THEN 13410
13400 IF (I - IEND) <= 0 THEN 13570
13410 XR = X: FR = F
13420 REM **********************************************
13430 REM TEST ON SATISFACTORY ACCURACY IN BISECTION LOOP
13440 REM **********************************************
13450 TOL = EPS
13460 IF (ABS(FR - FL) - TOL) <= 0 THEN 13530
13470 NEXT JK
13480 REM ************************************************************
13490 REM END OF BISECTION LOOP - NO CONVERGENCE AFTER IEND ITERATION STEPS
13500 REM FOLLOWED BY IEND SUCCESSIVE STEPS OF BISECTION
13510 REM ************************************************************
13520 IER = 1: GOTO 13780
13530 RETURN
13540 REM *************************************************************
13550 REM COMPUTATION OF ITERATED X-VALUE BY INVERSE PARABOLIC INTERPOLATION
13560 REM *************************************************************
```

```
13570 DA = FR - F: DX = (X - XL) * FL * (1 + F * (DA - TOL) / (DA * (FR - FL)))
/ TOL
13580 XM = X: FM = F: X = XL - DX: TOL = X
13590 GOSUB 12000
13600 F = TEMP: IF F = 0 THEN 13530
13610 REM ************************************************
13620 REM TEST ON SATISFACTORY ACCURACY IN ITERATION LOOP
13630 REM ************************************************
13640 TOL = EPS
13650 IF (ABS(F) - TOL) <= 0 THEN 13530
13660 REM **********************************
13670 REM PREPARATION OF NEXT BISECTION LOOP
13680 REM **********************************
13690 IF (SGN(F) + SGN(FL)) <> 0 THEN 13710
13700 XR = X: FR = F: GOTO 13260
13710 XL = X: FL = F: XR = XM: FR = FM: GOTO 13260
13720 REM ***************************************
13730 REM END OF ITERATION LOOP
13740 REM ERROR RETURN IN CASE OF WRONG INPUT DATA
13750 REM ***************************************
13760 IF XHI <> XLO THEN 13770 ELSE RETURN
13770 IER = 2: JK = 0
13780 BEEP: PRINT "NUMBER OF ITERATIONS= "; JK
13790 PRINT "ITERATION ROUTINE CONDITION CODE,IER= "; IER
13800 IF IER = 2 THEN PRINT "TCDR OUT OF TEMPERATURE RANGE"
13810 IF IER = 1 THEN PRINT "NO CONVERGENCE IN SUBROUTINE TRANS"
13820 STOP
14000 REM /////////////////////////////////////////////////////////////
14010 REM / SUBROUTINE GUESS TO DETERMINE INITIAL BOUNDS FOR ITERATION
14020 REM /////////////////////////////////////////////////////////////
14030 IF NFLAG = 0 THEN XLO = TAMB: XHI = 1000: DIV = 10
14040 IF NFLAG = 1 THEN XLO = 0: XHI = 10 * AT: DIV = 10
14050 CHA = (XHI - XLO) / DIV: NUM = INT(DIV): X = XLO
14060 GOSUB 12000
14070 FO = TEMP
14080 FOR JK = 1 TO NUM
14090 X = XLO + JK * CHA: GOSUB 12000
14100 FF = TEMP: IF (SGN(FF) + SGN(FO)) = 0 THEN 14140
14110 FO = FF
14120 NEXT JK
14130 XLI = XLO: XRI = XHI: RETURN
14140 XRI = X: XLI = X - CHA: RETURN
15000 REM ///////////////////////////////////////////////////////////////////
15010 REM / SUBROUTINE TO CALCULATE THERMAL RATING GIVEN A CDR TEMP (TCDR),
15020 REM / AND CONDUCTOR PARAMETERS AND WEATHER CONDITIONS
15030 REM ///////////////////////////////////////////////////////////////////
15040 REM PRINT USING "TRYING A TCDR OF ####.### DEG C"; TCDR
15050 REM *******************************************************
15060 REM * CALC CONDUCTOR HEAT LOSS (QR) BY RADIATION (WATTS/M)
15070 REM *******************************************************
15080 T3 = TCDR + 273
15090 T4 = TAMB + 273
15102 QR = .0178 * EMISS * D * ((T3 / 100) ^ 4 - (T4 / 100) ^ 4)
15110 REM *****************************************************************
15120 REM * CALC CONDUCTOR HEAT LOSS BY CONVECTION (WATTS/M)
15125 REM * NOTE CONVECTION EQUATIONS FORM IS DIFFERENT THAN IN BODY OF 738
15128 REM * BUT THE RESULTS OF CALCULATION ARE THE SAME
15130 REM *****************************************************************
15140 T5 = (TCDR + TAMB) / 2
15160 U1 = 1.458E-06 * (T5 + 273) ^ 1.5 / (T5 + 383.4)
15172 P1 = (1.2932 - .0001525 * CDR.ELEV + 6.379E-09 * CDR.ELEV ^ 2) / (1 +
.00367 * T5)
15180 K1 = .02424 + 7.477E-05 * T5 - 4.407E-09 * T5 ^ 2
```

```
   IF DEBUG = 0 THEN PRINT #2, "U1,P1,K1 = "; U1, P1, K1
15182 REM ****************************************************************
15184 REM * CALC CONDUCTOR HEAT LOSS (QC) BY NATURAL CONVECTION (WATTS/M)
15186 REM ****************************************************************
15188 IF (TCDR - TAMB) < 0! THEN TCDR = TAMB + .1
15191 QC = .0205 * P1 ^ .5 * D ^ .75 * (TCDR - TAMB) ^ 1.25
15192 IF VWIND = 0 THEN 15450
15194 REM ***************************************************************
15196 REM * CALC CONDUCTOR HEAT LOSS (QCF) BY FORCED CONVECTION (WATTS/M)
15198 REM ***************************************************************
15202 Z = D * P1 * VWIND / U1
15212 Q1 = .0119 * Z ^ .6 * K1 * (TCDR - TAMB)
15222 Q2 = (1.01 + .0372 * Z ^ .52) * K1 * (TCDR - TAMB)
15230 IF Q1 - Q2 <= 0 THEN 15260
15240 QCF = Q1
15250 GOTO 15270
15260 QCF = Q2
15265
15270 QCF = QCF * YC
15370 REM *********************************************************
15380 REM * SELECT LARGER OF CONVECTIVE HEAT LOSSES (QC VERSUS QCF)
15390 REM *********************************************************
15400 IF QCF < QC THEN 15450
15410 QC = QCF
15420 REM ************************************
15430 REM * CALC SUM OF STEADY STATE HEAT FLOWS
15440 REM ************************************
15450 R5 = -QS + QC + QR
15460 REM ***********************************************
15470 REM * CALC SQRT OF CONDUCTOR RESISTANCE IN OHMS/M
15480 REM ***********************************************
15492 W4 = SQR(B1 + B * TCDR)
15500 IF R5 <= 0 THEN TR = 0: GOTO 15560
15510 R4 = R5 ^ .5
15520 REM *************************************************
15530 REM * CALCULATE THERMAL RATING (AMPACITY) IN AMPERES
15540 REM *************************************************
15550 TR = R4 / W4
15560 RETURN
20000 REM /////////////////////
20010 REM / COMMENTS ON PROGRAM
20020 REM /////////////////////
20030 REM *
20040 REM * THE PROGRAM DOES NOT CALCULATE ANY INTERNAL RADIAL OR AXIAL
20050 REM * TEMPERATURE GRADIENTS. THIS IS NORMALLY NOT A SOURCE OF
20060 REM * SIGNIFICANT ERROR EXCEPT FOR INTERNALLY COMPLEX CONDUCTORS
20070 REM * SUCH AS FIBER-OPTIC SHIELD WIRE AND FOR NON-HOMOGENEOUS CONDUCTORS
20080 REM * FOR FAULT CURRENTS OF LESS THAN 1 MINUTE. THE PROGRAM DOES NOT
20090 REM * APPLY TO INTERNALLY COMPLEX CONDUCTORS, IT DOES CALCULATE A WORST
20100 REM * CASE ESTIMATE OF TEMPERATURE/RATING FOR ACSR OR ACSR/AW BY
NEGLECTING
20110 REM * THE HEAT STORAGE CAPACITY OF THE RELATIVELY POORLY CONDUCTING CORE
20120 REM * FOR STEP CURRENTS WHICH PERSIST FOR LESS THAN ONE MINUTE.
20130 REM * THE VARIATION IN SPECIFIC HEAT WITH TEMPERATURE IS NEGLECTED.
20140 REM * ADDED COMMENTS 7/97 DAD
20150 REM * ADDED SI FORMULAS, SOLAR EQUATIONS, AND CHANGED AIR PARAMETERS
```

Annex B

(informative)

Numerical example: steady-state conductor temperature calculation (SI units)

Input File - Steady State Conductor Temperature Calculation - SI Units

```
1                "NSELECT VALUE"
1000               "STEADY STATE CURRENT IN AMPERES"
40                "AMBIENT TEMP IN DEG C IS"
0.61               "WIND SPEED (M/SEC)"
90                "ANGLE BETWEEN WIND & CDR AXIS IN DEG"
"400 MM2 DRAKE 26/7 ACSR"  "CONDUCTOR DESCRIPTION"
28.12              "CONDUCTOR DIAMETER (MM)"
25,75              "MIN & MAX CDR TEMP IN DEG C"
0.07284, 0.08689         "MIN & MAX CDR RAC (OHMS/KM)"
0.5, 0.5            "COEF OF EMISS AND SOLAR ABSORP"
0.              "CDR ELEV ABOVE SEA LEVEL IN METERS"
45                "CDR DIRECTION CW RELATIVE TO NORTH"
43                "CDR LATITUDE IN DEGREES"
12, 161     "SOLAR HOUR 14 = 2PM OR 99(NO SUN) & DAY OF THE YEAR"
0, "CLEAR"          "AIR CLARITY - CLEAR(0), INDUST(1)"
```

Output File - Steady State Conductor Temperature Calculation - SI Units

```
--------------------------------------------------------
IEEE STD 738-2012 METHOD FOR CALCULATION OF
BARE OVERHEAD CONDUCTOR TEMPERATURES & THERMAL RATINGS
STDY STATE CDR TEMP CALC
400 MM2 DRAKE 26/7 ACSR
AIR TEMPERATURE IS 40 DEG C
WIND SPEED IS .61 M/SEC
ANGLE BETWEEN WIND AND CONDUCTOR IS 90 DEG
COEFFICIENT OF EMISSIVITY IS .5
COEFFICIENT OF ABSORPTIVITY IS .5
LINE DIRECTION IS 45 DEG FROM NORTH AND THE ATMOSPHERE IS CLEAR

STEADY STATE THERMAL CALCULATIONS
QS IS  13.738 WATTS PER METER OF CONDUCTOR
R IS  24.791 WATTS PER METER OF CONDUCTOR
QC IS  83.061 WATTS PER METER OF CONDUCTOR

GIVEN A CONSTANT CURRENT OF 1000 AMPERES
THE CONDUCTOR TEMPERATURE IS 100.7 DEG C
```

Annex C

(informative)

Numerical example: Steady-state thermal rating calculation (SI units)

Input File - Steady State Conductor Thermal Rating Calculation - SI Units

```
2               "NSELECT"
101.1             "STEADY STATE CURRENT "
40               "AMBIENT TEMP IN DEG C IS"
0.61              "WIND SPEED (M/SEC)"
90               "ANGLE BETWEEN WIND & CDR AXIS IN DEG"
"400 MM2 DRAKE 26/7 ACSR"  "CONDUCTOR DESCRIPTION"
28.12             "CONDUCTOR DIAMETER (MM)"
25,75             "MIN & MAX CDR TEMP IN DEG C"
0.07284, 0.08689        "MIN & MAX CDR RAC (OHMS/KM)"
0.5, 0.5            "COEF OF EMISS AND SOLAR ABSORP"
0.               "CDR ELEV ABOVE SEA LEVEL IN METERS"
45       "CDR DIRECTION IN DEG CW FROM NORTH"
43               "CDR LATITUDE IN DEGREES"
12, 161            "SOLAR HOUR 2PM=14 OR 99(NO SUN) & DAY OF THE YEAR"
0 , "CLEAR"          "AIR CLARITY - CLEAR(0), INDUST(1)"
```

Output File - Steady State Conductor Thermal Rating Calculation - SI Units

```
--------------------------------------------------------
IEEE STD 738-2012 METHOD FOR CALCULATION OF
BARE OVERHEAD CONDUCTOR TEMPERATURES & THERMAL RATINGS
STDY STATE RATING CALC
400 MM2 DRAKE 26/7 ACSR
AIR TEMPERATURE IS 40 DEG C
WIND SPEED IS .61 M/SEC
ANGLE BETWEEN WIND AND CONDUCTOR IS 90 DEG
COEFFICIENT OF EMISSIVITY IS .5
COEFFICIENT OF ABSORPTIVITY IS .5
LINE DIRECTION IS 45 DEG FROM NORTH AND THE ATMOSPHERE IS CLEAR


STEADY STATE THERMAL CALCULATIONS
QS IS  13.732 WATTS PER METER OF CONDUCTOR
QR IS  24.998 WATTS PER METER OF CONDUCTOR
QC IS  83.600 WATTS PER METER OF CONDUCTOR

GIVEN A MAXIMUM CONDUCTOR TEMPERATURE OF 101.1 DEG C,
THE STEADY STATE THERMAL RATING IS 1003 AMPERES
```

Annex D

(informative)

Numerical calculation of transient conductor temperature

Input File - Transient Conductor Temperature Calculation - SI Units

```
3               "NSELECT"
1               "PRE-STEP AMP(1) OR TEMP(2)"
400              "PRE-STEP CURRENT"
1200              "POST-STEP CURRENT"
1               "UNITS OF STEP MIN(1) OR SEC(0)"
15               "STEP DURATION IN ABOVE UNITS"
60               "CALC TIME INTERVAL (SEC)"
40               "AMBIENT TEMP IN DEG C IS"
0.61              "WIND SPEED (M/SEC)"
90               "ANGLE BETWEEN WIND & CDR AXIS IN DEG"
"400 MM2 DRAKE 26/7 ACSR"  "CONDUCTOR DESCRIPTION"
28.12             "CONDUCTOR DIAMETER (MM)"
25,75             "MIN & MAX CDR TEMP IN DEG C"
0.07284, 0.08689       "MIN & MAX CDR RAC (OHMS/KM)"
2               "CONDUCTOR HAS A STEEL CORE"
1066., 243           "HEAT CAP OF OUTER & CORE STRANDS (W-SEC/M-C)"
0.5, 0.5            "COEF OF EMISS AND SOLAR ABSORP"
0.         "CDR ELEV ABOVE SEA LEVEL IN METERS"
45        "CDR DIR IN DEG CW FROM NORTH"
43               "CDR LATITUDE IN DEGREES"
12, 161     "SOLAR HOUR (2PM=14) OR 99(NO SUN) & DAY OF THE YEAR"
0, "CLEAR"           "AIR CLARITY - CLEAR(0), INDUST(1)"
```

Output File - Transient Conductor Temperature Calculation - SI Units

```
-----------------------------------------------------------
IEEE STD 738-2012 METHOD FOR CALCULATION OF
BARE OVERHEAD CONDUCTOR TEMPERATURES & THERMAL RATINGS
CDR TEMP VS TIME CALC
400 MM2 DRAKE 26/7 ACSR
AIR TEMPERATURE IS 40 DEG C
WIND SPEED IS .61 M/SEC
ANGLE BETWEEN WIND AND CONDUCTOR IS 90 DEG
COEFFICIENT OF EMISSIVITY IS .5
COEFFICIENT OF ABSORPTIVITY IS .5
LINE DIRECTION IS 45 DEG FROM NORTH AND THE ATMOSPHERE IS CLEAR


   TRANSIENT THERMAL CALCULATIONS
INITIAL STEADY STATE CDR TEMP IS 55.7 DEG C
FOR A GIVEN INITIAL CURRENT OF  400 AMPERES,
       CORE HEAT CAPACITY = 243.0 WATTS-SEC/M-C
  OUTER STRAND LAYERS HEAT CAPACITY = 1066.0 WATTS-SEC/M-C
THE MAXIMUM TIME OF INTEREST AFTER THE STEP CURRENT
INCREASES TO  1200.0 AMPS IS  15.0000 MINUTES
THE MAX ALLOWABLE CONDUCTOR TEMPERATURE IS 1000.0 DEG C
TIME= 0.000 MIN  CDRTEMP= 55.7 DEG C
TIME= 1.000 MIN  CDRTEMP= 60.5 DEG C
TIME= 2.000 MIN  CDRTEMP= 65.0 DEG C
```

```
TIME= 3.000 MIN  CDRTEMP= 69.2 DEG C
TIME= 4.000 MIN  CDRTEMP= 73.2 DEG C
TIME= 5.000 MIN  CDRTEMP= 76.9 DEG C
TIME= 6.000 MIN  CDRTEMP= 80.3 DEG C
TIME= 7.000 MIN  CDRTEMP= 83.6 DEG C
TIME= 8.000 MIN  CDRTEMP= 86.6 DEG C
TIME= 9.000 MIN  CDRTEMP= 89.5 DEG C
TIME= 10.000 MIN  CDRTEMP= 92.1 DEG C
TIME= 11.000 MIN  CDRTEMP= 94.6 DEG C
TIME= 12.000 MIN  CDRTEMP= 96.9 DEG C
TIME= 13.000 MIN  CDRTEMP= 99.0 DEG C
TIME= 14.000 MIN  CDRTEMP= 101.1 DEG C
TIME= 15.000 MIN  CDRTEMP= 102.9 DEG C
-----------------------------------------------------------
```

Annex E

(informative)

Numerical Calculation of Transient Thermal Rating

```
Input File - Transient Conductor Thermal Rating Calculation - SI Units

4               "NSELECT"
2               "PRE-STEP AMP(1) OR TEMP(2)"
40               "PRE-STEP CONDUCTOR TEMPERATURE"
150              "MAX ALLOWED POST-STEP CDR TEMP"
1               "UNITS OF TIME STEP MIN(1) OR SEC(0)"
15               "STEP DURATION TO REACH MAX CDR TEMP"
60               "CALC TIME INTERVAL (SEC)"
40               "AMBIENT TEMP IN DEG C IS"
0.61              "WIND SPEED (M/SEC)"
90               "ANGLE BETWEEN WIND & CDR AXIS IN DEG"
"400 MM2 DRAKE 26/7 ACSR"  "CONDUCTOR DESCRIPTION"
28.12             "CONDUCTOR DIAMETER (MM)"
25,75             "MIN & MAX CDR TEMP IN DEG C"
0.07284, 0.08689       "MIN & MAX CDR RAC (OHMS/KM)"
2               "CONDUCTOR HAS A STEEL CORE"
1066., 243.0         "HEAT CAP OF OUTER & CORE STRANDS (W-SEC/M-C)"
0.5, 0.5           "COEF OF EMISS AND SOLAR ABSORP"
0.               "CDR ELEV ABOVE SEA LEVEL IN METERS"
45        "CDR DIRECTION IN DEG CW FROM NORTH"
43               "CDR LATITUDE IN DEGREES"
12, 161           "SOLAR HR 14=2PM OR 99(NO SUN) & DAY OF THE YEAR"
0, "CLEAR"           "AIR CLARITY - CLEAR(0), INDUST(1)"


Output File - Transient Conductor Thermal Rating Calculation - SI Units

-----------------------------------------------------------
IEEE STD 738-2012 METHOD FOR CALCULATION OF
BARE OVERHEAD CONDUCTOR TEMPERATURES & THERMAL RATINGS
TRANSIENT RATING CALC
400 MM2 DRAKE 26/7 ACSR
AIR TEMPERATURE IS 40 DEG C
WIND SPEED IS .61 M/SEC
ANGLE BETWEEN WIND AND CONDUCTOR IS 90 DEG
COEFFICIENT OF EMISSIVITY IS .5
COEFFICIENT OF ABSORPTIVITY IS .5
LINE DIRECTION IS 45 DEG FROM NORTH AND THE ATMOSPHERE IS CLEAR


   TRANSIENT THERMAL CALCULATIONS
INITIAL STEADY STATE CDR TEMP IS 40.0 DEG C

         CORE HEAT CAPACITY = 243.0 WATTS-SEC/M-C
  OUTER STRAND LAYERS HEAT CAPACITY = 1066.0 WATTS-SEC/M-C
THE MAX ALLOWABLE CONDUCTOR TEMPERATURE IS 150.0 DEG C
THE TRANSIENT THERMAL RATING IS   1642.0 AMPERES
THAT IS, WITH THIS CURRENT, THE CONDUCTOR TEMPERATURE JUST REACHES
THE MAXIMUM ALLOWABLE CDR TEMP OF 150.0 DEG C
IN  15.00 MINUTES
```

Annex F

(informative)

Conductor thermal time constant

The non-steady-heat balance Equation (2) cannot be solved analytically for conductor temperature as a function of time since certain of its terms are non-linear. Considering the equation term by term, it may be seen that the Joule heating term and the forced convection equation term are linear in conductor temperature. The solar heating term is also linear since it is independent of conductor temperature. The radiation heat loss term and the natural convection (zero wind speed) term are both non-linear in conductor temperature.

Several references [B5], [B20], and [B26] describe methods of approximating the radiation cooling Equation (7) as a linear function of temperature. Doing so yields a linear non-steady-state heat balance equation of the form:

$$\frac{d}{dt}(T_c - T_a) = K_1(T_c - T_a) + K_2 I^2 \tag{F1}$$

For a step change in electrical current, the solution of the linearized non-steady-state heat balance equation is:

$$T_c(t) = T_i + (T_f - T_i) \cdot (1 - e^{-t/\tau}) \tag{F2}$$

The steady state conductor temperature prior to the step increase in current is T_i. The steady-state conductor temperature which occurs long after the step increase in current is T_f. The thermal time constant, τ, may be calculated by use of the formula:

$$\tau = \frac{(T_f - T_i) \cdot mC_p}{R(T_c) \cdot (I_f^2 - I_i^2)} \tag{F3}$$

where the conductor resistance is that corresponding to the average conductor temperature, $(T_i + T_f)/2$.

Consider the exponential change in conductor temperature shown in Figure F.1. This is the "1200 amp" curve shown previously in Figure 2. The initial conductor temperature is 80 °C. The final conductor temperature is 128 °C. The current undergoes a step change from 800 A to 1200 A. If the average conductor temperature is taken as 100 °C, the resistance of the "Drake" conductor is 2.86×10^{-5} Ω/ft. From 5.6 of this standard, the heat capacity of the conductor is 399 W-s/ft-°C. The time constant is:

$$\tau = \frac{(128 - 80) \cdot 399}{2.86 \cdot 10^{-5} (1200^2 - 800^2)} = 837\ s$$

$$= 14\ \text{min}$$

Alternatively, the temperature change reaches 63% of its final value at a conductor temperature of 80 °C + (128 – 80) · 63 = 110 °C. In Figure F.1, this corresponds to a time of about 13 min.



Figure F.1 – Conductor temperature vs. time curve

Annex G

(informative)

Selection of weather conditions

In a recently published *CIGRE Technical Brochure 299*, "Guide for Selection of Weather Parameters for Bare Overhead Conductor Ratings" [B8], the issue of selecting weather conditions for line ratings is discussed at some length based on an exhaustive review of the technical literature. The recommendations provided in [B8] represent a practical guide for developing conservative thermal rating estimates for overhead lines, assuming that the engineer recognizes the need for normal clearance and safety margins employed in the design and operation of overhead transmission lines. There were also a series of qualifying objectives set forth by the brochure for the determination of the weather parameters.

The default recommendations for base ratings in [B8] are based on the likelihood of coincident worst rating conditions. For example, while effective wind speeds are sometimes lower than 0.6 m/s, it is extremely unlikely that they are lower than that value if the ambient temperature and solar radiation are high. Nevertheless, if, for example, ambient temperature and solar radiation are lower, wind speeds lower than 0.6 m/s are more likely. Similarly, values of total solar radiation (the sum of direct, indirect, and reflected radiation) can be higher or lower than 1000 W/m^2 varying with time-of-day and season and reflected radiation, caused by ground albedo, can be negligible or as large as 15% to 25% of direct radiation in the visual and 25% to 35% in the near-infrared range. In [B8] it was decided to simply use a solar heat intensity of 1000 W/m^2 direct radiation as a part of most severe coincident conditions (In IEEE Std 738-2012, indirect and reflected solar radiation are ignored but the direct solar heat intensity can be calculated as a function of date, time-of-day, and latitude).

In short summary, selection of ratings can be chosen using four different levels: base ratings; study-based ratings; ambient-adjusted ratings; or real time ratings.

G.1 Base ratings

Base ratings may be applied for any transmission line and should be used unless the utility adopts alternative practices as described below.

a) For sag-limited lines, [B8] recommends that base ratings be calculated for an effective wind speed of 0.6 m/s (2 ft/s), an ambient temperature close to the annual maximum of ambient temperature along the line route and a solar radiation of 1000 W/m^2 (92.9 W/ft^2). When combined with an assumed conductor absorptivity of no less than 0.8 and emissivity of no more than 0.1 below absorptivity, this combination can be considered safe for thermal rating calculations without field measurements.

b) For those lines where annealing of conductors is the primary concern, having narrow, sheltered corridors, with energized conductors either below tree canopy height or between buildings, the base rating should be estimated based on either a 0.4 m/s (1.3 ft/s) effective wind speed or by reducing the maximum conductor design temperature by 10 °C. Although the average conductor temperature, which determines the line sag, is not likely to be higher than that based on 0.6 m/s (2 ft/s) wind speed, the local effective wind speed in sheltered locations may be significantly lower.

c) Seasonal ratings should be based on an ambient temperature close to the maximum value of the season along the line and other criteria above, although the precautions discussed in Section 4 of [B8] should be exercised.

G.2 Study-based ratings

The transmission line owner/operator may base the rating assumptions of selected lines or regions on actual weather or rating studies, provided that:

a) Rating weather studies are conducted in the actual transmission line environment, using the methods recommended in Section 5 of [B8]. If seasonal ratings are applied, such studies must include the respective seasons.

b) Alternatively, rating studies can be conducted with devices that monitor line tension, sag, clearance, or conductor temperature. The methods are specified in Section 5 of [B8].

G.3 Ambient-adjusted ratings

Ratings can be adjusted based on varying ambient temperatures measures at the time. These are termed continually ambient-adjusted ratings. In this case, unless real time rating systems are used, the wind speed should be based on the assumption of a more conservative effective wind speed than base ratings. The extensive literature review by the [B8] task force clearly indicates that ambient temperature and wind speed are not independent parameters, higher wind speeds being associated with high ambient temperatures.

If the base rating is to be adjusted for daytime conditions, [B8] recommends the following: if the ambient temperature adjustment is less than 8 °C compared to the temperature selected for base rating conditions (for example, if the base ambient temperature is 35 °C and the actual ambient temperature is between 35 °C and 27 °C), the effective wind speed should be selected as no higher than 0.5 m/s (1.64 ft/s). If the temperature adjustment is more than 8 °C, the effective wind speed should be selected as no more than 0.4 m/s (1.3 ft/s). For nighttime ambient-adjusted ratings (between sunset and sunrise when solar radiation is zero), wind speed should be selected as zero (natural convection only), and solar radiation can also be considered nil. Continually ambient-adjusted ratings can provide technically justified ampacity increases for lines that are designed for low maximum conductor temperatures (e.g., below 60 °C to 70 °C). On the other hand, they will generally not provide technically justified benefits for lines designed for 100 °C or higher temperatures and their use is not recommended.

If a study-based line rating is to be adjusted for ambient temperature, the engineer must be careful to reduce the assumed wind speed to account for correlation with ambient temperature. As with ambient adjustment of base ratings, the wind speed at night should be much lower.

G.4 Real-time ratings

Rather than using "worst-case" weather assumptions, the transmission line owner/operator may elect to use real time monitoring equipment for determining the line rating, provided:

a) Monitoring equipment meets the sensitivity, accuracy, and calibration requirements specified in Section 5 of [B8].

b) It has been verified that the lines that are to be monitored meet the design clearance requirements.

c) Monitors are installed in sufficient quantity to provide statistically valid information of the sag or temperature over the entire length of the monitored circuit. See Section 4.5 and Section 5.6 of [B8] for additional guidance.

d) The operator has the capability of adjusting the line current to the level of standard or enhanced ratings in emergency conditions.

Annex H

(informative)

Tables for solar heating and air properties

Algebraic equations are presented for air properties, solar angles, and heat flux in 4.5 of this standard. These equations are appropriate as part of the numerical calculation process that the standard recommends. Nonetheless, tables of some of these properties can provide valuable insight into the relationships of these properties.

Table H.1 – Viscosity, density, and thermal conductivity of air (SI)

Temperature T_{film}		Absolute or dynamic viscosity [B15] μ_f (kg/m-s)	Air density [B17] ρ_f (kg/m^3)				Thermal conductivity of air [B18] k_f (W/m–°C)
°F	°C		0 m	1000 m	2000 m	4000 m	
32	0	1.72e-05	1.293	1.147	1.014	0.785	0.0242
41	5	1.74e-05	1.270	1.126	0.995	0.771	0.0246
50	10	1.76e-05	1.247	1.106	0.978	0.757	0.0250
59	15	1.79e-05	1.226	1.087	0.961	0.744	0.0254
68	20	1.81e-05	1.205	1.068	0.944	0.731	0.0257
77	25	1.84e-05	1.184	1.051	0.928	0.719	0.0261
86	30	1.86e-05	1.165	1.033	0.913	0.707	0.0265
95	35	1.88e-05	1.146	1.016	0.898	0.696	0.0269
104	40	1.91e-05	1.127	1.000	0.884	0.685	0.0272
113	45	1.93e-05	1.110	0.984	0.870	0.674	0.0276
122	50	1.95e-05	1.093	0.969	0.856	0.663	0.0280
131	55	1.98e-05	1.076	0.954	0.843	0.653	0.0283
140	60	2.00e-05	1.060	0.940	0.831	0.643	0.0287
149	65	2.02e-05	1.044	0.926	0.818	0.634	0.0291
158	70	2.04e-05	1.029	0.912	0.806	0.625	0.0295
167	75	2.07e-05	1.014	0.899	0.795	0.616	0.0298
176	80	2.09e-05	1.000	0.887	0.783	0.607	0.0302
185	85	2.11e-05	0.986	0.874	0.773	0.598	0.0306
194	90	2.13e-05	0.972	0.862	0.762	0.590	0.0309
203	95	2.15e-05	0.959	0.850	0.752	0.582	0.0313
212	100	2.17e-05	0.946	0.839	0.741	0.574	0.0317

Table H.2 – Viscosity, density, and thermal conductivity of air (US)

Temperature T_{film}		Absolute or dynamic viscosity [B22] μ_f (lb/ft·hr)	Air density [B17] ρ_f (lb/ft^3)				Thermal conductivity of air [B25] k_f (W/ft-°C)
°F	°C		Sea level	5000 ft	10000 ft	15000 ft	
32	0	0.0415	0.0807	0.0671	0.0554	0.0455	0.00739
41	5	0.0421	0.0793	0.0660	0.0545	0.0447	0.00750
50	10	0.0427	0.0779	0.0648	0.0535	0.0439	0.00762
59	15	0.0433	0.0765	0.0636	0.0526	0.0431	0.00773
68	20	0.0439	0.0752	0.0626	0.0517	0.0424	0.00784
77	25	0.0444	0.0740	0.0616	0.0508	0.0417	0.00795
86	30	0.0450	0.0728	0.0606	0.0500	0.0411	0.00807
95	35	0.0456	0.0716	0.0596	0.0492	0.0404	0.00818
104	40	0.0461	0.0704	0.0586	0.0484	0.0397	0.00830
113	45	0.0467	0.0693	0.0577	0.0476	0.0391	0.00841
122	50	0.0473	0.0683	0.0568	0.0469	0.0385	0.00852
131	55	0.0478	0.0672	0.0559	0.0462	0.0379	0.00864
140	60	0.0484	0.0661	0.0550	0.0454	0.0373	0.00875
149	65	0.0489	0.0652	0.0542	0.0448	0.0367	0.00886
158	70	0.0494	0.0643	0.0535	0.0442	0.0363	0.00898
167	75	0.0500	0.0634	0.0527	0.0436	0.0358	0.00909
176	80	0.0505	0.0627	0.0522	0.0431	0.0354	0.00921
185	85	0.0510	0.0616	0.0513	0.0423	0.0347	0.00932
194	90	0.0515	0.0608	0.0506	0.0418	0.0343	0.00943
203	95	0.0521	0.0599	0.0498	0.0412	0.0338	0.00952
212	100	0.0526	0.0591	0.0492	0.0406	0.0333	0.00966

Table H.3 is provided for the user's convenience. It lists values of solar attitude and azimuth as a function of latitude for the day of the year that yields peak solar heating.

Table H.3 – Altitude, H_c, and Azimuth, Z_c, in degrees of the sun at various latitudes for an annual peak solar heat input

Degrees north latitude	Local sun time						
Lat	10:00AM		Noon		2:00PM		
	H_c	Z_c	H_c	Z_c	H_c	Z_c	N
–80	32	33	33	180	32	327	350
–70	40	37	43	180	40	323	350
–60	48	43	53	180	48	317	350
–50	55	52	63	180	55	308	350
–40	60	66	73	180	60	294	350
–30	62	83	83	180	62	277	350
–20	62	96	90	180	62	264	20
–10	61	97	88	180	61	263	50
0	60	91	90	180	60	269	80
+10	61	85	89	180	61	275	110
20	62	85	90	180	62	275	140
30	62	97	83	180	62	263	170
40	60	114	73	180	60	245	170
50	55	128	63	180	55	232	170
60	48	137	53	180	48	223	170
70	40	143	43	180	40	217	170
80	32	147	33	180	32	213	170

Table H.4 lists values of total heat flux received by a surface at sea level (heat intensity), as a function of the solar altitude, H_c, for two levels of atmospheric clarity. It is included for the user's convenience.

Table H.4 – Total heat flux received by a surface at sea level normal to the sun's rays as a function of solar altitude

Degrees solar altitude, H_c	Clear atmosphere		Industrial atmosphere	
	Q_S (W/m^2)	Q_S (W/ft^2)	Q_S (W/m^2)	Q_S (W/ft^2)
5	234	21.7	136	12.6
10	433	40.2	240	22.3
15	583	54.2	328	30.5
20	693	64.4	422	39.2
25	770	71.5	502	46.6
30	829	77.0	571	53.0
35	877	81.5	619	57.5
40	913	84.8	662	61.5
45	941	87.4	694	64.5
50	969	90.0	727	67.5
60	1000	92.9	771	71.6
70	1020	95.0	809	75.2
80	1030	95.8	833	77.4
90	1040	96.4	849	78.9

Table H.5— Solar heat multiplying factors, K_{solar} for high altitudes [B16]

Elevation above sea level H_e - m	Multiplier for values in Table H.4	Elevation above sea level H_e - ft	Multiplier for values in Table H.4
0	1.00	0	1.00
1 000	1.10	5 000	1.15
2 000	1.19	10 000	1.25
4 000	1.28	15 000	1.30